BREV

Display

(HP-41CX, Hewlett Packard 1983 and DM41X, SwissMicros 2020)

Overview[1]

Programs BREVB and BREVD are binary reversi games. The user must flip the bits such that all 9 bits will be 0 whilst applying the least number of flips. Although both programs do the same, the implementations differ.

Program BREVB does a binary flip of each bit. That means, if the value is 0 make it 1 and if it is 1 make it 0. This is done in the loop around LBL 04. This version does not use a decimal separator.

Program BREVD does a decimal flip of each bit and uses a decimal separator for each group of 3 bits. It adds 1's to the bits [ii] in the original number [i] and truncates the 2's (sum of a bit with value 1 and the mask with value 1). Here is an example with number of flips equals 6 with a given pattern [i] 100.111.001:

1. to swap the last 6 bits, the user enters the value 6 which will generate a value of six 1's, in the loop around LBL 01, giving:

   111.111 [ii]

2. this flip value (ii) will be added to the decimal patterns, i.e.:

```
100.111.001   [i]
    111.111   [ii]
------------ +
100.222.112   [iii]
```

3. locate where the 2's are and generate a subtract value, which in the case above is:

   222.002 [iv]

4. subtract the updated 2's mask (iv) from (iii) which will give the new (i)':

```
100.222.112   [iii]
    222.002   [iv]
------------ -
100.000.110   [i]'
```

5. the game ends when [i]' is 0.

The display uses flag 29 to show the thousands separator. However, it should not show the last decimal separator when "1E 9" is there to keep the leading zeros. This is done by adding three alpha characters,

[1] *This program is copyright and is supplied without representation or warranty of any kind. The author assumes no responsibility and shall have no liability, consequential or otherwise, of any kind arising from the use of this program material or any part thereof*

rotating them including the decimal separator and remove the first 6 alpha characters from the alpha register, in LBL 05, specifically lines 32-40.

In program BREVD the random number generator checks for each of the nine 10^X values whether it should be added and delivers a number between 0-111.111.1111.

Both programs carry out a game-over test is all bits have the value 0.

Program BREVB is more efficient with handling the bits during the generation of the pattern as well as flipping the bits. It does not provide decimal separators for ease of use.

Flipping bits is counted from right to left.

Example (1): BREVB

KEYSTROKES	DISPLAY	COMMENTS
[XEQ] [ALPHA] BREVB [ALPHA]	00 1 100 1 10:_:0	Start; counter is 0 (most right)
1 [R/S]	00 1 100 1 1 C:_:1	Flip last bit
3 [R/S]	00 1 100000:_:2	Flip last 3 bits
5 [R/S]	00 1 1 1 1 1 C:_:3	Flip last 5 bits
7 [R/S]	0000000000:_:4	Flip last 7 bits; game over. Flag 0 set
[R/S]	0,000000	Finish game
[R/S]	0 1 10 1 100 C:_:0	Start new game
8 [R/S]	000 100 1 10:_:1	Try left to right, flip last 8 bits
6 [R/S]	0000 1 100 C:_:2	Flip last 6 bits
5 [R/S]	000000 1 10:_:3	Flip last 5 bits
3 [R/S]	0000000000 C:_:4	Flip last 3 bits
1 [R/S]	0000000000:_:5	Flip last bit; game over. Flag 0 set
[R/S]	0,000000	End game

Example (2): BREVD

KEYSTROKES	DISPLAY	COMMENTS
[XEQ] [ALPHA] BREVD [ALPHA]	100.0 1 1.100:_:0	Start; counter is 0 (most right)
2 [R/S]	100.0 1 1.1 1 C:_:1	Flip last 2 bits
5 [R/S]	100.000.000:_:2	Flip last 5 bits
9 [R/S]	0 1 1.1 1 1.1 1 C:_:3	Flip last 9 bits
8 [R/S]	0000000000:_:4	Flip last 8 bits; game over. Flag 0 set
[R/S]	-4,000000	Finish game
[R/S]	00 1.100.00 C:_:0	Start new game
7 [R/S]	000.0 1 1.1 10:_:1	Try left to right, flip last 7 bits
5 [R/S]	000.000.00 C:_:2	Flip last 5 bits
1 [R/S]	0000000000:_:3	Flip last bit; game over. Flag 0 set
[R/S]	-4,000000	End game

Program Listing

The listing of the programs is given below with 2 XROM functions SEED and RNDM on lines 6 and 9 resp. 9 and 15. These can be taken as explicit calls to other programs in memory or to XROM functions, for example the CCD module. If replaced by these XROM functions (as in the RAW and TXT file) the total number of byte savings is 8. So, program BREVB would be 94 instead of 102 bytes and program BREVD would be 162 instead of 170 bytes. If not available, programs must be in place for these functions.

The listing of program BREVB ("binary" flip) is given below:

```
01▪LBL "BREVB"
02 FIX 0
03 CF 29
04 CLX
05 STO 00
06 XEQ "SEED"
07 9
08▪LBL 01
09 XEQ "RNDM"
10 RND
11 STO IND Y
12 RDN
13 DSE X
14 GTO 01
15▪LBL 03
16 CLA
17 SF 00
18 SF 01
19 9
20▪LBL 02
21 1
22 X#NN?
23 CF 01
24 CLX
25 X#NN?
26 CF 00
27 RDN
28 ARCL IND X
29 DSE X
30 GTO 02
31 >":_:"
32 ARCL 00
33 PROMPT
34 FS?C 00
35 SF 01
36 FS?C 01
37 GTO 05
38 ISG 00
39▪LBL 04
40 1
41 X=NN?
42 CLX
43 STO IND Y
44 RDN
45 DSE X
46 GTO 04
47 GTO 03
48▪LBL 05
49 FIX 5
50 SF 29
51 END
```

(102 bytes)

and for BREVD ("decimal" flip) shown here:

```
01▪LBL "BREVD"
02 9
03 FIX 0
04 CF 29
05 CF 00
06 CF 01
07 ,
08 STO 00
09 XEQ "SEED"
10▪LBL 04
11 RCL Y
12 1
13 -
14 10^X
15 XEQ "RNDM"
16 RND
17 *
18 +
19 DSE Y
10 GTO 04
21▪LBL 05
22 STO 01
23 X=0?
24 SF 00
25 9 E-9
26 1/X
27 INT
28 -
29 X=0?
30 SF 01
31 RCL 01
32 1E 9
33 +
34 CLA
35 SF 29
36 ARCL X
37 >"123"
38 -4
39 AROT
40 ASHF
41 >":_:"
42 CF 29
43 ARCL 00
44 PROMPT
45 FS?C 00
46 SF 01
47 FS?C 01
48 GTO 06
49 ,
50 ISG 00
51▪LBL 01
52 RCL Y
53 1
54 -
55 10^X
56 +
57 DSE Y
58 GTO 01
59 ST+ 01
60 ,
61 STO 02
62 RCL 01
63▪LBL 02
64 10
65 /
66 STO Z
67 FRC
68 ,2
69 -
70 X<>0?
71 GTO 03
72 RDN
73 10^X
74 ST+ 02
75 LASTX
76 X<>Y
77▪LBL 03
78 1
79 ST+ Z
70 RDN
81 X<> Z
82 INT
83 X<>0?
84 GTO 02
85 RCL 01
86 RCL 02
87 ST+ X
88 -
89 GTO 05
90▪LBL 06
91 FIX 5
92 SF 29
```

(170 bytes)

Registers, Labels and Flags

REGISTERS	COMMENTS
R00	Counter number of flips
R01	Value [i] (BREVD only)
R02	Mask 2's [iv] (BREVD only)
R01-R09	Pattern bits (BREVB only)

LABELS BREVB	COMMENTS
LBL01	Randomise value [i]
LBL02	Check if bits are 0
LBL03	Show bits for display
LBL04	Flip bits
LBL05	Reset display settings
LABELS BREVD	
LBL01	Generate [ii] + [iii]
LBL02	Make [iv]
LBL03	Make [iv]
LBL04	Randomise pattern value
LBL05	Check bits and display
LBL06	Reset display settings

FLAGS	COMMENTS
00	Set if all bits are 0
01	Set to support game end

Downloads

The RAW/TXT format of the program is available via the website: BREV (in zip file).